Computation of the Regular Confluent Hypergeometric Function

Julio Abad and Javier Sesma, Universidad de Zaragoza

A procedure, alternative to `Hypergeometric1F1`, for the computation of the regular confluent hypergeometric function $_1F_1(a;b;z)$ is suggested. The procedure, based on an expansion of the Whittaker function in series of Bessel functions, proves to be useful for large values of $|az|$, whenever $|z|$ is smaller than or comparable to 1.

The numerical values of the confluent hypergeometric (Kummer) function, $_1F_1(a;b;z)$ or $M(a, b, z)$, obtained by `Hypergeometric1F1[a,b,z]` are correct for moderate values of the parameters a and b and the variable z. However, if the parameter a is large, the function loses accuracy and computation time increases, unless the variable z is small enough that $|az|$ is less than or comparable to 1. In this note, we propose a procedure to evaluate $_1F_1(a;b;z)$ when the values of the parameters and the variable are unfavorable for using `Hypergeometric1F1[a,b,z]`. The procedure is based on an expansion, given by Buchholz [Buchholz 1969, sec. 7.4], of the Whittaker function in terms of Bessel functions. The Kummer function, which is closely related to the Whittaker function, has an expansion

$$_1F_1(a;b;z) = \Gamma(b)\mathrm{e}^{z/2}2^{b-1}\sum_{n=0}^{\infty} p_n(b,z)\frac{J_{b-1+n}(\sqrt{z(2b-4a)})}{(\sqrt{z(2b-4a)})^{b-1+n}} \quad (1)$$

where $p_n(b, z)$ are polynomials in b and z. These polynomials were introduced by Buchholz and are described below.

Another expansion similar to this one could also be used [Abramowitz and Stegun 1965, eq. 13.3.7; Luke 1969, sec. 4.8]:

$$_1F_1(a;b;z) = \Gamma(b)\mathrm{e}^{z/2}2^{b-1}\sum_{n=0}^{\infty} A_n z^n \frac{J_{b-1+n}(\sqrt{z(2b-4a)})}{(\sqrt{z(2b-4a)})^{b-1+n}}$$

$$A_0 = 1, \qquad A_1 = 0, \qquad A_2 = b/2 \quad (2)$$
$$nA_n = (n-2+b)A_{n-2} + (2a-b)A_{n-3}$$

The equivalence of these two expansions can be checked by using the recurrence relations satisfied by the Bessel functions [Abramowitz and Stegun 1965, eq. 9.1.27]. At first sight, expansion (2) appears easier to implement. However, it has the drawback that the coefficients A_n depend on both a and b, and, for large a, the A_n increase with n roughly as $a^{n/3}$. The coefficients $p_n(b, z)$ of expansion (1), instead, do not depend on a, and their dependence on b and z can be separated, as shown below. In addition, since the numerical coefficients of the powers of b and z in $p_n(b, z)$ do not depend on a, b, or z, they can be tabulated in the case of frequent evaluation of $_1F_1$. These coefficients are rational and can be handled by *Mathematica* with infinite precision.

Julio Abad received his Ph.D. in Physics from the University of Zaragoza in 1973*, where he is now Professor of Theoretical Physics. His interests include elementary particle physics and integrable spin systems.*

Javier Sesma obtained his Ph.D. in 1964 *at the University of Barcelona. He has taught at the Universities of Barcelona, Caracas, Valencia, and Zaragoza, where he is now Professor of Theoretical Physics. His main interest is in special func-*

Buchholz Polynomials

The polynomials $p_n(b, z)$ are defined by the closed contour integral

$$p_n(b,z) = \frac{(iz)^n}{2\pi i}\int^{(0+)} \exp\left(\frac{iz}{2}\left(\cot v - \frac{1}{v}\right)\right)\left(\frac{\sin v}{v}\right)^{b-2}\frac{1}{v^{n+1}}\,dv$$

or, equivalently, by

$$p_n(b,z) = \frac{(iz)^n}{n!}\lim_{v\to 0}\frac{d^n}{dv^n}\left(\exp\left(\frac{iz}{2}\left(\cot v - \frac{1}{v}\right)\right)\left(\frac{\sin v}{v}\right)^{b-2}\right)$$

To implement its computation in *Mathematica*, we have obtained an expression of $p_n(b, z)$ as a sum of products of polynomials in b and in z, separately. These polynomials can be obtained recursively, as we will show.

Let us define the functions

$$F(b,v) \equiv \left(\frac{\sin v}{v}\right)^{b-2}$$
$$G(z,v) \equiv \exp\left(\frac{iz}{2}\left(\cot v - \frac{1}{v}\right)\right)$$

Then,

$$p_n(b,z) = \frac{(iz)^n}{n!} \lim_{v\to 0} \sum_{m=0}^{n} \binom{n}{m} \frac{d^{n-m}G(z,v)}{dv^{n-m}} \frac{d^m F(b,v)}{dv^m} \tag{3}$$

On the other hand, by denoting $H(v) \equiv \cot v - 1/v$, we have

$$\begin{aligned} \frac{dF(b,v)}{dv} &= (b-2)H(v)F(b,v) \\ \frac{dG(z,v)}{dv} &= \frac{iz}{2}\frac{dH(v)}{dv}G(z,v) \end{aligned}$$

and, by succesive derivation,

$$\begin{aligned} \frac{d^k F(b,v)}{dv^k} &= (b-2)\sum_{r=0}^{k-1}\binom{k-1}{r}\frac{d^{k-1-r}H(v)}{dv^{k-1-r}}\frac{d^r F(b,v)}{dv^r} \\ \frac{d^m G(z,v)}{dv^m} &= \frac{iz}{2}\sum_{j=0}^{m-1}\binom{m-1}{j}\frac{d^{m-j}H(v)}{dv^{m-j}}\frac{d^j G(z,v)}{dv^j} \end{aligned}$$

Taking the limit $v \to 0$ in these equations and introducing the notation

$$\begin{aligned} F_k(b) &\equiv \lim_{v\to 0}\frac{d^k F(b,v)}{dv^k} \\ g_m(z) &\equiv \lim_{v\to 0}\frac{d^m G(z,v)}{dv^m} \\ h_r &\equiv \lim_{v\to 0}\frac{d^r H(v)}{dv^r} \end{aligned}$$

we obtain

$$\begin{aligned} F_0(b) &= 1 \\ F_k(b) &= (b-2)\sum_{r=0}^{k-1}\binom{k-1}{r}h_{k-1-r}F_r(b), \quad k=1,2,\ldots \end{aligned} \tag{4}$$

$$\begin{aligned} g_0(z) &= 1 \\ g_m(z) &= \frac{iz}{2}\sum_{j=0}^{m-1}\binom{m-1}{j}h_{m-j}g_j(z), \quad m=1,2,\ldots \end{aligned} \tag{5}$$

The values h_m are immediately obtained from the series expansion [Abramowitz and Stegun 1965, eq. 4.3.70]

$$H(v) = -\sum_{k=1}^{\infty}\frac{2^{2k}|B_{2k}|}{(2k)!}v^{2k-1}$$

where B_{2k} are the Bernoulli numbers [Abramowitz and Stegun 1965, table 23.2]. Obviously,

$$h_{2k}=0, \quad h_{2k+1} = -\frac{2^{2k+1}|B_{2k+2}|}{k+1}, \quad k=0,1,2,\ldots \tag{6}$$

The first of these equations implies, with equation 4, that $F_{2k+1}(b) \equiv 0$. By denoting $f_s(b) \equiv F_{2s}(b)$ and using equation 6 in equations 4 and 5, one obtains for the polynomials $f_s(b)$ and $g_m(z)$ the recurrences

$$\begin{aligned} f_0(b) &= 1 \\ f_s(b) &= -\left(\frac{b}{2}-1\right)\sum_{r=0}^{s-1}\binom{2s-1}{2r}\frac{4^{s-r}|B_{2(s-r)}|}{s-r}f_r(b), \\ &\qquad s=1,2,\ldots \end{aligned}$$

$$\begin{aligned} g_0(z) &= 1 \\ g_m(z) &= -\frac{iz}{4}\sum_{k=0}^{[\frac{m-1}{2}]}\binom{m-1}{2k}\frac{4^{k+1}|B_{2(k+1)}|}{k+1}g_{m-2k-1}(z), \\ &\qquad m=1,2,\ldots \end{aligned}$$

The Buchholz polynomials are then obtained from equation 3, which, with the introduced notation, reads

$$p_n(b,z) = \frac{(iz)^n}{n!}\sum_{s=0}^{[\frac{n}{2}]}\binom{n}{2s}f_s(b)g_{n-2s}(z)$$

Here are *Mathematica* functions that compute the Buchholz polynomials $p_n(b, z)$.

```
Buchholzf[0, b_] = 1;

Buchholzf[s_, b_] := -(b/2 - 1) *
  Sum[Binomial[2s-1, 2r] 4^(s-r)/(s-r) *
      Abs[BernoulliB[2(s-r)]] Buchholzf[r, b],
    {r, 0, s-1}]

Buchholzg[0, z_] = 1;

Buchholzg[m_, z_] := -I z/4 *
  Sum[Binomial[m-1, 2k] 4^(k+1)/(k+1) *
      Abs[BernoulliB[2(k+1)]] Buchholzg[m-2k-1, z],
    {k, 0, (m-1)/2}]

BuchholzP[n_Integer?NonNegative, b_?NumberQ, z_?NumberQ] :=
(I z)^n/n! *
  Sum[Binomial[n, 2s] Buchholzf[s, b] *
      Buchholzg[n-2s, z],
        {s, 0, n/2}]
```

An Alternative to Hypergeometric1F1

The known relation between the Bessel function and the hypergeometric function ${}_0F_1$ [Abramowitz and Stegun 1965, eq. 9.1.69] can be used to write the expansion (1) in the form

$${}_1F_1(a;b;z) = e^{z/2}\sum_{n=0}^{\infty}p_n(b,z)\frac{{}_0F_1(b+n;t)}{2^n(b)_n} \tag{7}$$

where $t \equiv z(a-b/2)$ and the Pochhammer symbol is defined by

$$(b)_0 = 1, \quad (b)_n = b(b+1)\ldots(b+n-1), \quad n=1,2,3,\ldots$$

For large values of $|az|$, this expansion has the advantage of converging much more rapidly than the conventional series expansion of ${}_1F_1$. Its implementation in *Mathematica*, with the sum truncated at $n = m$, is immediate:



FIGURE 1. The number of significant terms in the expansion of ${}_1F_1(a;b;z)$, as a function of a in the upper-half complex plane, with $b = 1$, $z = 1$, and a `$MachinePrecision` of 16.



FIGURE 2. The number of significant terms in the expansion of ${}_1F_1(a;b;z)$ as a function of a and z, with $b = 1$ and a `$MachinePrecision` of 16.

```
Hypergeometric1F1Buchholz[a_?NumberQ, b_?NumberQ,
        z_?NumberQ, m_Integer?Positive] :=
  E^(z/2) Sum[BuchholzP[j, b, z] *
    Hypergeometric0F1[b+j, z(a-b/2)]/(2^j Pochhammer[b, j]),
  {j, 0, m}]
```

The number of terms to be taken in the sum in the right hand side of equation 7 (the parameter m in `Hypergeometric1F1Buchholz[a,b,z,m]`) should be adjusted to the required precision. For larger values of $|a|$, or smaller values of $|z|$, fewer terms are needed to obtain the same precision. Figure 1 shows the number m of significant terms in the expansion as a function of a in the upper-half complex plane, for $b = 1$ and $z = 1$, and a `$MachinePrecision` of 16. Figure 2 shows the number of significant terms as a function of a and z, for $b = 1$.

Terms in the expansion corresponding to an index $n \geq m$ have an absolute value less than 10^{-16} times the absolute value of the sum of the preceding terms. Here is a procedure to compute ${}_1F_1$ that automatically truncates the sum, to avoid computing terms with relative values less than `$MachineEpsilon`:

```
Hypergeometric1F1Buchholz[a_?NumberQ, b_?NumberQ,
        z_?NumberQ] :=
Module[{sum, term, nmax = 20}, Catch[
  sum = Hypergeometric0F1[b, z(a-b/2)];
  Do[
    term =
      BuchholzP[n, b, z] Hypergeometric0F1[b+n, z(a-b/2)]/
            (2^n Pochhammer[b, n]);
    sum = sum + term;
    If[Abs[N[term]] < Abs[N[sum]] $MachineEpsilon,
        Throw[sum E^(z/2)]],
    {n, nmax}];
  Message[Hypergeometric1F1Buchholz::ncvi];
  sum E^(z/2) ] ]
```

A comparison of the values obtained with the conventional and the alternative procedures can be seen in Table 1. The timings were made with *Mathematica* Version 2.2. Obviously, the alternative procedure becomes more advantageous as the modulus of the parameter a increases.

a	**conventional**	**timing**	**alternate**	**timing**
10^3	$4.4234\ 10^{22}$	0.1	$4.4234\ 10^{22}$	1.1
-10^3	$-113216.$	0.0	$-1.00968\ 10^{-7}$	2.2
10^4	$4.55306\ 10^{84}$	1.8	$4.55306\ 10^{84}$	0.4
-10^4	$-7.14658\ 10^{66}$	0.0	$-2.73909\ 10^{-11}$	0.4
10^5	$1.9064\ 10^{287}$	18.7	$1.9064\ 10^{287}$	0.3
-10^5	$-7.26065\ 10^{269}$	0.0	$-1.68809\ 10^{-14}$	0.3
10^6	$5.549127\ldots\ 10^{934}$	194.7	$5.549127\ldots\ 10^{934}$	0.2
-10^6	$-3.9\ 10^{913}$	0.9	$2.67216\ 10^{-17}$	0.2

TABLE 1. Values of ${}_1F_1(a;b;z)$ obtained with the built–in function `Hypergeometric1F1` and the alternate procedure `Hypergeometric1F1Buchholz`, for different values of a and fixed $b = 6.8$, $z = 1.2$. The timings are given in seconds.

References

Abramowitz, M., and I. A. Stegun, eds. 1965. *Handbook of Mathematical Functions*. Dover, New York.

Buchholz, H. 1969. *The Confluent Hypergeometric Function*. Springer-Verlag, Heidelberg.

Luke, Y. L. 1969. *The Special Functions and their Approximations*. Academic Press, New York.

Julio Abad and Javier Sesma
Departamento de Fisica Teorica,
Facultad de Ciencias, E-50009 Zaragoza, Spain
julio@cc.unizar.es

The electronic supplement contains the package `Buchholz.m`.